

Schnelle Frequenzgangberechnung in ANSYS / HF

Ausgabe: 5 / 2003

Problem:

Bei Simulationen in der Hochfrequenztechnik interessieren oft nicht nur die S-Parameter der Bauteile bei einer einzigen Frequenz, sondern in einem weiten Frequenzbereich. Bisher war es notwendig, für jeden Frequenzpunkt eine harmonische Berechnung durchzuführen, was hohe Rechenzeiten bedeutete.

Lösung:

Durch Verwenden des neuen ANSYS Series Xpansion Frequency Sweep Modules (FS Module, ist gesondert zu lizenzieren) können nun in einer einzigen Analyse die S-Parameter für den gesamten Frequenzbereich berechnet werden.

Beispiel:

Es soll untersucht werden, inwieweit eine Verletzung des Dielektrikums eines Koaxialkabels Auswirkungen auf das Übertragungsverhalten hat. Das Kabel wird im Frequenzbereich von 200 bis 600 MHz untersucht und der Transmissionskoeffizient S_{21} für 41 Frequenzen berechnet und dargestellt.

Das Dielektrikum des Koaxialkabels ist in einem Bereich unterbrochen ($\epsilon_r = 1$), Aussen- und Innenleiter aber noch intakt.

Es werden zwei HF-Ports modelliert, von denen einer als Anregeport einer TEM Welle definiert wird (HFPORT, ..., COAX, ..., TEM, SOFT, ..., 1), der andere als S-Parameter-Messport (HFPORT, ..., COAX, ..., TEM, SEXT, ..., 0). Um das Kabel an den Enden reflexionsfrei abzuschließen, sind perfekt absorbierende Schichten (Perfectly Matched Layers, PML) eingefügt. Innenleiter und Abschirmung werden als perfekt elektrisch leitend angenommen (AX=0).

Über den Parameter SWPOPT wird gesteuert, ob eine komplette harmonische Analyse oder der Fast Frequency Sweep durchgeführt wird. Die Rechnung über die Frequenzen wird mit dem Befehl **SPSWP**, *Anfangsfrequenz*,

Endfrequenz, *Frequenzinkrement*, *Sweeptyp*, ..., *Ausgabeformat* gestartet.


Sweeptyp=0 bedeutet Fast Frequency Sweep, 1 voll harmonischer Sweep.

Ausgabeformat steuert die Ausgabe in eine Datei im Touchstone Format (*file.s2p*), im Beispiel wird 1 für Ausgabe der S-Parameter in Betrag (dB) und

Phase (Winkelgrad) gewählt. Der Befehl **plsp**, *Ausgabeformat*, *Portj*, *Portj*, *Portk*, ... plottet die S-Parameter S_{jj} , S_{jk} , ... über der Frequenz.

Die Kurven zeigen eine gute Übereinstimmung, die Abweichungen sind kleiner als 0.05dB. Die Rechenzeit verkürzt sich bei Anwendung des FFS um fast 50% (69sek gegenüber 128sek). Je mehr Frequenzpunkte gerechnet werden sollen, desto größer ist der Rechenzeitgewinn.

Vergleich Fast Frequency Sweep -
Full Harmonic Sweep



Schnelle Frequenzgangberechnung in ANSYS / HF

Ausgabe: 5 / 2003

Input:


```
swpopt=0
d_i=0.75e-3 ! Innendurchmesser
d_a=4.8e-3 ! Aussendurchmesser
! Bestimmen von eps aus der Geometrie
eps=(log(d_a/d_i)*60/75)**2
pi=acos(-1)
c0=2.9979e8
c=c0/sqrt(eps)
lambda=c/f_frq
l_k=1*lambda
l_ls=.25*lambda
ndf=1.0
/prep7
cyl4,,,5*d_i,0,.5*d_a,90
et,1,120
et,2,120,,,1 ! PML Region
et,9,200,7 ! Hilfsmesh
type,9
lsel,s,length,,,25*pi*d_i
asll,s
lsla,s
lesiz,all,,,5*ndf
mat,2
amap,1,1,2,3,4
alls

/com, Leiter
type,1
mat,2
extopt,esize,10*ndf, 1/((l_ls/5)/(0.5*l_k)*10*ndf)
vext,all,,,,.5*l_k

/com, Spalt
asel,s,loc,z,.5*l_k
mat,3
extopt,esize,5*ndf,1
vext,all,,,,l_ls

/com, Leiterfortsetzung
asel,s,loc,z,.999*(.5*l_k+l_ls),1.001*(.5*l_k+l_ls)
mat,2
extopt,esize,10*ndf, ((l_ls/5)/(0.5*l_k)*10*ndf)
vext,all,,,,.5*l_k

alls
acle,all
etdele,9
```



```
/com, Uebergangs und PML Regionen
local,11,0,,,,,90
local,12,0,,,,,-90
csys,0
asel,s,loc,z,0
mat,2
type,1
esys,0
extopt,esize,5*ndf
vext,all,,,,-.5*lambda
asel,s,loc,z,0-.5*lambda
mat,2
type,2
esys,11
vext,all,,,,-.5*1*lambda
asel,s,loc,z,l_k+l_ls
mat,2
type,1
esys,0
extopt,esize,5*ndf
vext,all,,,,.5*1*lambda
asel,s,loc,z,l_k+l_ls+.5*1*lambda
mat,2
type,2
esys,12
vext,all,,,,.5*1*lambda
esys,0
```

Schnelle Frequenzgangberechnung in ANSYS / HF

Ausgabe: 5 / 2003

Input:

```
pmlopt,11,one,1e-4 ! Daempfung PML
pmlopt,12,one,1e-4

mp,perx,1,1 ! Materialien
mp,murx,1,1
mp,perx,2,eps
mp,murx,2,1
mp,perx,3,1
mp,murx,3,1

! Zuweisung der HF-Ports
lsel,s,loc,z,0
lsel,r,length,,.5*(d_a-d_i)
asll,s
asel,r,loc,z,0
bfa,all,port,1
hfport,1,coax,0,tem,soft,.5*d_i,.5*d_a,1,0
lsel,s,loc,z,l_k+l_ls
lsel,r,length,,.5*(d_a-d_i)
asll,s
asel,r,loc,z,l_k+l_ls
bfa,all,port,2
hfport,2,coax,0,tem,sext,.5*d_i,.5*d_a,0,0

! Leitfähigkeiten sind ideal bei In.- und Aussenleiter
csys,1
alls
asel,s,loc,x,.49*d_a,.51*d_a ! Abschirmung
lsel,s,length,,.5*(d_a-d_i)
lsel,r,loc,z,.5*l_k,.5*l_k+l_ls
lsel,a,length,,l_ls
lsel,a,length,,.25*pi*d_a
asel,a,ext ! Rechenrand+Innenleiter
asel,u,loc,y,0,1 ! ohne Symmetrieränder
asel,u,loc,y,89,91
da,all,ax,0
fini
```

```
/solu
alls
*if,swpopt,eq,1,then
spswp,200e6, f_frq,10e6,1,,1! FULL HARMONIC
*elseif,swpopt,eq,0,then
spswp,200e6, f_frq,10e6,0,,1!SERIES XPANSION
*endif
fini
plsp,1,1,2 ! Ausgabe in dB, S21
```

